# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: kirek <rbuldo@gmail.com>
# Contributor: Cédric Bellegarde <gnumdk@gmail.com>

pkgname=lollypop
pkgver=1.4.37
pkgrel=3
pkgdesc='Music player for GNOME'
arch=(any)
url=https://gitlab.gnome.org/World/lollypop
license=(GPL)
depends=(
  dconf
  gdk-pixbuf2
  gobject-introspection-runtime
  gst-plugins-base-libs
  gst-python
  gstreamer
  gtk3
  hicolor-icon-theme
  libhandy
  libsoup3
  pango
  python
  python-beautifulsoup4
  python-cairo
  python-gobject
  python-pillow
  totem-plparser
)
makedepends=(
  appstream-glib
  git
  gobject-introspection
  intltool
  itstool
  meson
)
optdepends=(
  'easytag: Modify tags'
  'gst-libav: FFmpeg plugin for GStreamer'
  'gst-plugins-bad: "Bad" plugin libraries'
  'gst-plugins-base: "Base" plugin libraries'
  'gst-plugins-good: "Good" plugin libraries'
  'gst-plugins-ugly: "Ugly" plugin libraries'
  'kid3-qt: Store covers in tags'
  'libsecret: Last.FM support'
  'youtube-dl: Youtube support'
)
_tag=847c7640b1e1b82170dd8bd81651d518e3446bd4
source=(git+https://gitlab.gnome.org/World/lollypop.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd lollypop
  git describe --tags
}

build() {
  arch-meson lollypop build \
    --libexecdir=lib/lollypop
  meson compile -C build
}

package() {
  DESTDIR="${pkgdir}" meson install -C build
}

# vim: ts=2 sw=2 et:
